

# Hands on activities 5

## - Interrupt and Timer

**DR. B. I. MORSHED**

ASSOCIATE PROFESSOR  
COMPUTER SCIENCE  
TEXAS TECH UNIVERSITY

---

**Embedded Systems/Cyber Physical Systems**

**CS 4380 / CS 5331**

# Example 1: PCINT interrupt

---

```
#include <avr/interrupt.h>
#include <avr/sleep.h>

void setup() {
  cli(); // Clear global interrupt

  // Set Pin 13 as output and 10 as input
  DDRB |= B100000 ;
  DDRB &= B111011 ;

  // Control regs for PCINT
  PCICR |= B00000001 ; // Enable PCINT0
  PCMSK0 |= B00000100 ; // Select PCINT0 mask

  // Serial.begin(9600); // Only for debug

  // Use an appropriate sleep mode
  set_sleep_mode(SLEEP_MODE_PWR_DOWN);

  sei(); // Set global Interrupt
}


// ISR for pin change interrupt capture
// Note: triggers both on rising & falling
ISR(PCINT0_vect) {
  // Display in serial monitor for debug
  // Serial.println("Switch pressed");

  // Toggle the LED
  PORTB ^= B100000 ;
}


// Main loop
void loop() {
  // Display in serial monitor for debug
  // Serial.println("Main loop");

  // Do nothing!
  // Put MCU to sleep
  sleep_mode();
}
```

## Hardware setup:



- From Vdd (5 V), connect a push switch and a 1k $\Omega$  resistor in series to ground (0V)
- Pin 10 connects to the midpoint of the push switch and 1k $\Omega$  resistor
- From pin 13, connect an LED (red) with a 220 $\Omega$  (or 330 $\Omega$ ) resistor in series to ground



# Example 2: Timer Interrupt

---

```
#include <avr/interrupt.h>
#include <avr/sleep.h>
void setup () {
    cli(); // Disable global interrupt - atomic
    DDRB |= B100000; // Pin 13 output
    // Set timer 1 to normal mode
    TCCR1A = B00000000;
    // Set pre-scaler to 1024
    TCCR1B = B00000101;
    // Turn ON OVF
    TIMSK1 = B00000001;
    // Initial Timer1 value for 1 sec count
    TCNT1 = 0xC2F6;

    // Use an appropriate sleep mode
    set_sleep_mode(SLEEP_MODE_IDLE);

    sei(); // Enable global interrupt
}
```

```
// Timer1 ISR
ISR(TIMER1_OVF_vect) {
    // Toggle output pin each 1 sec
    PORTB ^= B100000;
    // Reset counter value for next
    TCNT1 = 0xC2F6;
}

// Main loop
void loop () {
    // Do nothing
    // If including sleep mode, ensure
    // timer1 is ON while sleep
    sleep_mode();
}
```

---

### **Hardware connection:**

- Connect pin 13 to an LED in series of 220  $\Omega$  or 330  $\Omega$  resistor to ground.

### **Calculation:**

Internal clock = 16 MHz

As the needed time is large, use maximum pre-scaler of 1024

Clock cycles needed =  $16 * 10^6 / 1024 = 15,625$

Timer1 is 16-bit, so maximum value (i.e. 0xFFFF) = 65,535

So, Timer1 count value needed =  $65,535 - 15,625 = 49,910$